

Ü b u n g s b l a t t 12

Wir bieten an, bearbeitete Aufgaben zu korrigieren, falls sie zum unten angegebenen Zeitpunkt abgeliefert werden. Mit \* und Punktangaben gekennzeichnete Aufgaben können dazu verwendet werden, Bonus-Punkte zu erwerben.

Bei der folgenden Aufgabe handelt es sich um ein individualisiertes „Miniprojekt“, in das die Matrikelnummer eingeht (jeder hat seine eigene Aufgabe). Es ist nur das gefragte Endergebnis auf dem beigefügten Formblatt abzugeben. Dementsprechend gibt es keinerlei Teilpunkte: das Ergebnis ist entweder 'richtig' oder 'falsch'.

**Achtung!** Aufgrund des nahenden Vorlesungsendes hat dieses Blatt eine Laufzeit von nur einer Woche!

**Aufgabe 47\*:** (Extrema mit Nebenbedingungen in der Ökonomie) (8 Punkte)

Auf einer soeben gekauften Produktionsmaschine kann ein Betrieb zwei unterschiedliche Produkte herstellen. Die Anschaffungskosten betragen 10 000 000 DM. Seien  $x_1, x_2$  die Anzahl der herzustellenden Einheiten von Produkt 1 bzw. Produkt 2. Es wird erwartet, daß die Maschine insgesamt

$$x_1 + x_2 = 1\,000\,000 \quad (\#)$$

Einheiten produzieren kann, bis sie erneuert werden muss. Die Produktionskosten pro Einheit betragen 10 DM beim ersten und 15 DM beim zweiten Produkt. Die Lagerhaltungskosten seien durch

$$300\,000 + \frac{100}{M} \cdot (x_1^2 + x_1 \cdot x_2 + x_2^2)$$

modelliert, wobei  $M$  die Matrikelnummer sei. Demgemäß sollen die Gesamtkosten als Funktion des 'Outputs'  $x_1, x_2$  durch

$$K(x_1, x_2) = 10\,300\,000 + 10 \cdot x_1 + 15 \cdot x_2 + \frac{100}{M} \cdot (x_1^2 + x_1 \cdot x_2 + x_2^2)$$

modelliert werden.

Die Marktpreise der Produkte sind  $p_1 = 50$  DM bzw.  $p_2 = 60$  DM pro Einheit. Die Maschine soll den Gewinn optimierend genutzt werden, d.h., der Gewinn

$$G(x_1, x_2) = p_1 \cdot x_1 + p_2 \cdot x_2 - K(x_1, x_2)$$

soll maximiert werden, wobei die Nebenbedingung (#) zu beachten ist.

Liefere den maximierten Gewinn bis auf  $\pm 1$  DM genau als Lösung auf dem beigefügten Formblatt ab!

**Musterlösung:**

Es ist der Gewinn

$$G(x_1, x_2) = 50 \cdot x_1 + 60 \cdot x_2 - 10\,300\,000 - 10 \cdot x_1 - 15 \cdot x_2 - \frac{100}{M} \cdot (x_1^2 + x_1 \cdot x_2 + x_2^2)$$

$$= 40 \cdot x_1 + 45 \cdot x_2 - 10\,300\,000 - \frac{100}{M} \cdot (x_1^2 + x_1 \cdot x_2 + x_2^2)$$

unter der Nebenbedingung

$$g(x_1, x_2) = x_1 + x_2 - 1\,000\,000 = 0$$

zu maximieren.

**Erster Weg:** Die Nebenbedingung hat eine derart einfache Form, daß man sie nach  $x_2$  auflösen kann:

$$x_2 = x_2(x_1) = 1\,000\,000 - x_1.$$

In den Gewinn eingesetzt wird dieser zu einer Funktion einer einzigen Variablen  $x_1$ , die wie üblich zu maximieren ist:

$$G(x_1, x_2(x_1)) = 40 \cdot x_1 + 45 \cdot x_2(x_1) - 10\,300\,000 - \frac{100}{M} \cdot (x_1^2 + x_1 \cdot x_2(x_1) + x_2^2(x_1)) =$$

$$40 \cdot x_1 + 45 \cdot x_2 \cdot (1\,000\,000 - x_1) - 10\,300\,000 - \frac{100}{M} \cdot (x_1^2 + x_1 \cdot (1\,000\,000 - x_1) + (1\,000\,000 - x_1)^2).$$

Z.B. für die Matrikelnummer  $M = 7654321$ :

```
>> M:= 6543210:
>> K:= (x1, x2) -> 10300000 + 10*x1 + 15*x2 + 100/M*(x1^2 + x1*x2 + x2^2):
>> G:= (x1, x2) -> 50*x1 + 60*x2 - K(x1, x2):
>> G(x1, x2)
```

$$40 \cdot x_1 + 45 \cdot x_2 - \frac{10 \cdot x_1 \cdot x_2}{654321} - \frac{10 \cdot x_1^2}{654321} - \frac{10 \cdot x_2^2}{654321} - 10300000$$

```
>> solve(diff(G(x1, 1000000 - x1), x1), x1)
```

```
{1345679/4}
```

```
>> float(%)
```

```
{336419.75}
```

```
>> x1:= 336419.75:
```

```
>> x2:= 1000000 - x1:
```

```
>> maxGewinn:= float(G(x1, x2))
```

```
21146686.69
```

**Zweiter Weg:** Man löst die Nebenbedingung nicht auf, sondern benutzt Lagrangesche Multiplikatoren. Man hat die Gleichungen

$$\frac{\partial G}{\partial x_1} = \lambda \cdot \frac{\partial g}{\partial x_1}, \quad \frac{\partial G}{\partial x_2} = \lambda \cdot \frac{\partial g}{\partial x_2},$$

zusammen mit der Nebenbedingung  $g(x_1, x_2) = 0$  nach  $x_1, x_2, \lambda$  aufzulösen:

```

>> delete x1, x2:
>> M:= 6543210:
>> Kosten:= (x1, x2) -> 10300000 + 10*x1 + 15*x2 + 100/M*(x1^2 + x1*x2 + x2^2):
>> Gewinn:= (x1, x2) -> 50*x1 + 60*x2 - K(x1, x2):
>> Nebenbedingung:= (x1, x2) -> x1 + x2 - 1000000:

```

Die zu maximierende Funktion `Gewinn` ist definiert, ebenso die `Nebenbedingung`. Die Methode der Lagrangeschen Multiplikatoren besagt, daß die 3 folgenden Gleichungen für  $x_1, x_2, \lambda$  zu lösen sind:

```

>> Gleichung1:= diff(Gewinn(x1, x2), x1) = lambda* diff(Nebenbedingung(x1, x2), x1)

```

$$40 - \frac{10 x_2}{654321} - \frac{20 x_1}{654321} = \text{lambda}$$

```

>> Gleichung2:= diff(Gewinn(x1, x2), x2) = lambda* diff(Nebenbedingung(x1, x2), x2)

```

$$45 - \frac{20 x_2}{654321} - \frac{10 x_1}{654321} = \text{lambda}$$

```

>> Gleichung3:= Nebenbedingung(x1, x2) = 0

```

$$x_1 + x_2 - 1000000 = 0$$

Dieses Gleichungssystem ist hier recht einfach: es sind 3 **lineare** Gleichungen.

```

>> numeric::solve([Gleichung1, Gleichung2, Gleichung3], [x1, x2, lambda])

```

```

{[x1 = 336419.75, x2 = 663580.25, lambda = 19.57547213]}

```

```

>> maxGewinn:= Gewinn(336419.75, 663580.25)

```

```

21146686.69

```

Für die Korrekture: folgende Prozedur berechnet für jede Matrikelnummer  $M$  den maximierten Gewinn:

```

maxGewinn:= proc(M)
local x1, x2, lambda,
    Lagerhaltung, Kosten, Gewinn,
    Nebenbedingung, Gleichung1,
    Gleichung2, Gleichung3, Loesung;
begin
    x1:= genident("x");
    x2:= genident("x");
    lambda:= genident("lambda");
    Lagerhaltung:= (x1, x2) -> 300000 + 100/M*(x1^2 + x1*x2 + x2^2);
    Kosten:= (x1, x2) -> 10000000 + 10*x1 + 15*x2 + Lagerhaltung(x1, x2);
    Gewinn:= (x1, x2) -> 50*x1 + 60*x2 - Kosten(x1, x2);
    //plotfunc3d(Gewinn(x1*10^6, x2*10^6)/10^8, x1 = 0..1, x2 = 0..1);
    Nebenbedingung:= (x1, x2) -> x1 + x2 - 1000000;
    Gleichung1:= diff(Gewinn(x1, x2), x1) = lambda*diff(Nebenbedingung(x1, x2), x1);
    Gleichung2:= diff(Gewinn(x1, x2), x2) = lambda*diff(Nebenbedingung(x1, x2), x2);

```

```
Gleichung3:= Nebenbedingung(x1, x2) = 0;
Loesung:= numeric::solve([Gleichung1, Gleichung2, Gleichung3], [x1, x2, lambda]);
subs(Gewinn(x1, x2), op(Loesung));
end_proc:

>> maxGewinn(6543210)
                21146686.69
```

---

ABGABE: Dienstag, 10.7.01, zu Beginn der Vorlesung.

Nur das beiliegende Formblatt abgeben, Name, Vorname, Matrikelnummer, Gruppe und Lösung leserlich(!) angeben.

**Nur dieses Formblatt abgeben!**

Mathe B für WiWi, Blatt 12, Abgabe: 10.7.01

Vorname:

Name:

Matrikelnummer (**unbedingt angeben!**):

Gruppe (**unbedingt ankreuzen!**):

- Di 14–16, P1510 Ettler (Gruppe 1)
- Di 16–18, D1.328 Koch (Gruppe 2)
- Do 9–11, P7203 Kaminski (Gruppe 7)
- Do 14–16, H4.113 Koch (Gruppe 4)
- Do 16–18, D1.338 Hufnagel (Gruppe 5)
- Fr 11–13, N4.325 Hufnagel (Gruppe 3)
- Fr 11–13, P1611 Morschel (Gruppe 6)
- Fr 11–13, D1.338 Kaminski (Gruppe 8)
- Fr 11–13, D1.328 Ettler (Gruppe 9)
- keine Gruppenzugehörigkeit

---

**Lösung der Aufgabe 47:** Der maximierte Gewinn ist

DM.

---

Aufgabe	47*
max Punkte	8
erreicht:	